
 

 

      ***************************************************************** 
      *                                                               * 
      *  WHEN KEYING A COBOL PROGRAM, THERE ARE TWO MARGINS THAT      * 
      *  REQUIRE CONSIDERATION.  THE 'A' MARGIN (COLUMNS 8 TO 11) AND * 
      *  THE 'B' MARGIN (COLUMNS 12 TO 72). ALL  DIVISION  NAMES,     * 
      *  SECTION NAMES, PARAGRAPH NAMES, FD'S, AND 01 AND 77 LEVELS   * 
      *  MUST BEGIN IN THE 'A' MARGIN, WHILE THE REST OF THE COBOL    * 
      *  STATEMENTS MUST BEGIN IN THE 'B' MARGIN. (IMPORTANT - DO NOT * 
      *  KEY PAST COLUMN 72 IN COBOL, BECAUSE THIS AREA IS RESERVED   * 
      *  FOR IDENTIFICATION OR SEQUENCING AND IS IGNORED BY THE       * 
      *  COMPILER.  THE BLANK LINES SEEN AT APPROPRIATE PLACES IN     * 
      *  THE  PROGRAM  ARE  INSERTED  BY  THE  PROGRAMMER TO  GREATLY * 
      *  IMPROVE PROGRAM READIBILITY. THEY  ARE  IGNORED BY THE       * 
      *  COMPILER, AND SHOULD BE USED IN ALL COBOL PROGRAMS FOR       * 
      *  READABILITY.                                                 * 
      *                                                               * 
      ***************************************************************** 
       IDENTIFICATION DIVISION. 
      ***************************************************************** 
      *                                                               * 
      *  IN THE IDENTIFICATION DIVISION,  THE ONLY REQUIRED STATEMENT * 
      *  IS PROGRAM-ID.  THE OTHERS  GIVE  PERTINANT INFORMATION      * 
      *  ABOUT THE PROGRAM.  THE REMARKS PARAGRAPH USUALLY CONTAINS A * 
      *  BRIEF DESCRIPTION OF THE PROGRAM, FILE INPUT, ETC.  THIS     * 
      *  PROVIDES A SIMPLE WAY TO FURTHER  DOCUMENT YOUR PROGRAM AND  * 
      *  WILL BE EXPECTED ON ALL ASSIGNMENTS.                         * 
      *                                                               * 
      ***************************************************************** 
 
       PROGRAM-ID. 
                         BILLING. 
       AUTHOR. 
                         GEORGE FOWLER-ED CHRISTENSON. 
       INSTALLATION. 
                         TEXAS A & M UNIVERSITY-CSUS. 
       DATE-WRITTEN. 
                         SEPTEMBER 8, 1985- JANUARY 31, 1991. 
       DATE-COMPILED. 
 
      / 
 
      ***************************************************************** 
      *                                                               * 
      *  ASSIGNMENT NUMBER                                            * 
      *  DATE ASSIGNED                                                * 
      *  DATE DUE                                                     * 
      *  PURPOSE:           THIS IS A BILLING PROGRAM FOR THE  CIRCLE * 
      *   CLUB RESTAURANT AND BAR. IT CALCULATES THE TOTAL AMOUNT DUE * 
      *   FROM EACH CUSTOMER AND A GRAND TOTAL DUE FROM ALL ACCOUNTS. * 
      *   THERE ARE DISCOUNTS ALLOWED UNDER CERTAIN CONDITIONS.  THEY * 
      *   ARE:                                                        * 
      *       2 PERCENT FOR A RESTAURANT BILL OVER $200.00           * 
      *       5 PERCENT FOR A BAR BILL OVER $400.00                   * 
      *       3 PERCENT FOR A TOTAL BILL OVER $1,000.00 AFTER         * 
      *         OTHER DISCOUNTS HAVE BEEN SUBTRACTED.                 * 
      *                                                               * 
      *   INPUT FILE SPECIFICATIONS:                                  * 



 

 

      *           COLUMNS                       DESCRIPTION           * 
      *             1- 3                        MEMBERSHIP NUMBER     * 
      *             5-29                        MEMBER NAME           * 
      *            31-48                        MEMBER ADDRESS        * 
      *            49-53                        RESTAURANT BILL       * 
      *            55-59                        BAR BILL              * 
      *            61-65                        ANY OTHER CHARGES     * 
      *            67-72                        PAST DUE AMOUNT       * 
      *                                                               * 
      *  THE OUTPUT CONTAINS MEMBERSHIP NUMBER,  NAME,  ADDRESS,  AND * 
      *  AMOUNT DUE.   THE SUMMARY CONTAINS THE TOTAL AMOUNT DUE FROM * 
      *  ALL CUSTOMERS.                                               * 
      *                                                               * 
      ***************************************************************** 
 
      / 
 
      ***************************************************************** 
      *                                                               * 
      *                  ENVIRONMENT DIVISION                         * 
      *                                                               * 
      ***************************************************************** 
       ENVIRONMENT DIVISION. 
 
      ***************************************************************** 
      *                                                               * 
      *  IN COBOL, ALL ASPECTS OF THE DATA  PROCESSING  PROBLEM  THAT * 
      *  DEPEND ON THE PHYSICAL CHARACTERISTICS OF A SPECIAL COMPUTER * 
      *  ARE WITHIN THE  ENVIRONMENT DIVISION  OF THE SOURCE  PROGRAM * 
      *  THUS A CHANGE IN  COMPUTERS  ENTAILS  MAJOR CHANGES  IN ONLY * 
      *  THE ENVIRONMENT DIVISION.      THE  ENVIRONMENT DIVISION  IS * 
      *  DIVIDED INTO TWO SECTIONS: THE CONFIGURATION SECTION AND THE * 
      *  INPUT-OUTPUT SECTION.  THE CONFIGURATION SECTION  HAS  THREE * 
      *  PARAGRAPHS WHICH WE WILL USE:   SOURCE-COMPUTER, OBJECT-COM- * 
      *  PUTER, AND SPECIAL-NAMES. THE INPUT-OUTPUT SECTION TELLS THE * 
      *  COMPUTER WHICH HARDWARE DEVICES ARE REQUIRED FOR I/O.        * 
      *                                                               * 
      ***************************************************************** 
 
       CONFIGURATION SECTION. 
 
       SOURCE-COMPUTER. 
                         PC. 
       OBJECT-COMPUTER. 
                         PC. 
                        
 
       INPUT-OUTPUT SECTION. 
 
       FILE-CONTROL. 
 
            SELECT CUSTOMER-FILE 
               ASSIGN TO "CUSTOMER.DAT". 
 
            SELECT REPORT-FILE 
                ASSIGN TO "BILLING.RPT". 
 



 

 

      / 
 
      ***************************************************************** 
      *                                                               * 
      *                       DATA DIVISION                           * 
      *                                                               * 
      ***************************************************************** 
       DATA DIVISION. 
 
      ***************************************************************** 
      *                                                               * 
      *  THE   DATA  DIVISION   OF  A  COBOL  PROGRAM  CONTAINS   THE * 
      *  DESCRIPTION OF ALL INFORMATION  TO BE PROCESSED BY  THE PRO- * 
      *  GRAM.    FIRST IS THE SECTION THAT CONTAINS A DESCRIPTION OF * 
      *  ALL EXTERNALLY STORED DATA.     THERE IS ALWAYS ONE FD (FILE * 
      *  DESCRIPTION)  FOR  EACH  SELECT  CLAUSE  IN THE INPUT-OUTPUT * 
      *  SECTION OF THE ENVIRONMENT DIVISION.    THE NEXT PART IS THE * 
      *  WORKING-STORAGE SECTION WHICH CONTAINS DESCRIPTIONS  OF  ALL * 
      *  RECORDS THAT ARE NOT PART OF EXTERNAL FILES,  BUT ARE DEVEL- * 
      *  OPED AND PROCESSED INTERNALLY.   EXAMPLES OF WORKING-STORAGE * 
      *  AREAS ARE  TOTALS  COUNTERS,  DESCRIPTIONS OF HEADINGS   AND * 
      *  DETAILED DESCRIPTIONS OF INPUT-OUTPUT RECORDS.               * 
      *                                                               * 
      ***************************************************************** 
 
      / 
 
       FILE SECTION. 
 
       FD  CUSTOMER-FILE 
           RECORD CONTAINS 72 CHARACTERS 
           BLOCK CONTAINS 88 RECORDS 
           LABEL RECORDS ARE OMITTED 
           DATA RECORD IS CUSTOMER-RECORD. 
 
       01  CUSTOMER-RECORD             PIC X(72). 
 
       FD  REPORT-FILE 
           LABEL RECORDS ARE OMITTED 
           DATA RECORD IS REPORT-LINE. 
 
       01  REPORT-LINE         PIC X(132). 
 
      / 
 
      ***************************************************************** 
      *                                                               * 
      *                   WORKING STORAGE SECTION                     * 
      *                                                               * 
      ***************************************************************** 
       WORKING-STORAGE SECTION. 
 
       01  WS1-ACCUMULATORS-COUNTERS. 
           05  WS1-TOTAL-BILL               PIC 9(6)V99  VALUE ZERO. 
           05  WS1-TOTAL-AMT-DUE            PIC 9(6)V99  VALUE ZERO USAGE COMP. 
           05  WS1-PAGE-NUM                 PIC 9(8)     VALUE ZERO USAGE COMP. 
 



 

 

       01  WS2-INDICATORS-GROUP. 
           05  WS2-ARE-THERE-MORE-RECORDS   PIC X(3)     VALUE 'YES'. 
               88 WS2-MORE-RECORDS                       VALUE 'YES'. 
               88 WS2-NO-MORE-RECORDS                    VALUE 'NO '. 
 
       01  WS3-CUSTOMER-RECORD-IN. 
           05  WS3-MEMBER-NUMBER-IN          PIC 9(3). 
           05                                PIC X(1). 
           05  WS3-NAME-IN                   PIC X(25). 
           05                                PIC X(1). 
           05  WS3-MEMBER-ADDRESS-IN         PIC X(18). 
           05  WS3-REST-BILL                 PIC 9(3)V99. 
           05                                PIC X(1). 
           05  WS3-BAR-BILL                  PIC 9(3)V99. 
           05                                PIC X(1). 
           05  WS3-OTHER-CHARGES             PIC 9(3)V99. 
           05                                PIC X(1). 
           05  WS3-AMT-OUTSTANDING           PIC 9(4)V99. 
           05                                PIC X(8). 
 
      /     
 
       01  WS4-DETAIL-LINE. 
           05                                PIC X(22)  VALUE SPACES. 
           05  WS4-MEMBER-NUMBER-OUT         PIC 9(3). 
           05                                PIC X(19)  VALUE SPACES. 
           05  WS4-NAME-OUT                  PIC X(26). 
           05                                PIC X(10)  VALUE SPACES. 
           05  WS4-MEMBER-ADDRESS-OUT        PIC X(19). 
           05                                PIC X(13)  VALUE SPACES. 
           05  WS4-AMT-DUE                   PIC $ZZ,ZZ9.99. 
           05                                PIC X(11)  VALUE SPACES. 
 
       01  WS5-TITLE. 
           05                                PIC X(61)  VALUE SPACES. 
           05                                PIC X(11)  VALUE 
                    'CIRCLE CLUB'. 
           05                                PIC X(61)  VALUE SPACES. 
 
 
       01  WS6-COLUMN-HEADINGS. 
           05                                PIC X(15)  VALUE SPACES. 
           05                                PIC X(17)  VALUE 
                    'MEMBERSHIP NUMBER'. 
           05                                PIC X(18)  VALUE SPACES. 
           05                                PIC X(4)   VALUE 'NAME'. 
           05                                PIC X(30)  VALUE SPACES. 
           05                                PIC X(7)   VALUE 'ADDRESS'. 
           05                                PIC X(22)  VALUE SPACES. 
           05                                PIC X(10)  VALUE 
                    'AMOUNT DUE'. 
           05                                PIC X(10)  VALUE SPACES. 
 
       01  WS7-TOTAL-LINE. 
           05                                PIC X(36)  VALUE SPACES. 
           05                                PIC X(43)  VALUE 
                    'THE TOTAL AMOUNT DUE FROM ALL CUSTOMERS IS'. 



 

 

           05  WS7-TOTAL-AMOUNT              PIC $ZZZ,ZZ9.99. 
           05                                PIC X(43)  VALUE SPACES. 
 
       01  WS8-BLANK-LINE                    PIC X(132) VALUE SPACES. 
 
      /          
 
       01  WS9-STAR-LINE. 
           05                                PIC X(9)   VALUE SPACES. 
           05                                PIC X(114) VALUE ALL '*'. 
           05                                PIC X(10)  VALUE SPACES. 
 
       01  WS10-CURRENT-DATE. 
           05  WS10-YEAR-IN                  PIC X(2). 
           05  WS10-MONTH-IN                 PIC X(2).  
           05  WS10-DAY-IN                   PIC X(2). 
 
       01  WS11-DATE-LINE. 
           05                                PIC X(3)   VALUE SPACES. 
           05                                PIC X(6)   VALUE 'DATE: '. 
           05  WS11-DATE-OUT. 
               10  WS11-MONTH-OUT            PIC X(2). 
               10  WS11-DAY-OUT              PIC /X(2)/. 
               10  WS11-YEAR-PUT             PIC X(2). 
           05                                PIC X(106) VALUE SPACES. 
           05                                PIC X(6)   VALUE 'PAGE: '. 
           05  WS11-PAGE-NO-OUT              PIC ZZ. 
           05                                PIC X(2)   VALUE SPACES. 
 
      / 
 
      ***************************************************************** 
      *                                                               * 
      *                     PROCEDURE DIVISION                        * 
      *                                                               * 
      ***************************************************************** 
 
       PROCEDURE DIVISION. 
 
      ****************************************************************** 
      *                                                                * 
      * THE PROCEDURE DIVISION OF A COBOL PROGRAM CONTAINS THE ACTUAL  * 
      * INSTRUCTIONS FOR SOLVING A PROBLEM  WHICH MAY INCLUDE          * 
      * SENTENCES, STATEMENTS, CONDITIONAL STATEMENTS, PARAGRAPHS,     * 
      * PROCEDURES AND SECTIONS.                                       * 
      *                                                                * 
      ****************************************************************** 
 
 
       0000-MAIN-DRIVER. 
            PERFORM A000-INITIALIZATION. 
            PERFORM B000-CLUB-CUSTOMER-PROCESS 
                    UNTIL WS2-NO-MORE-RECORDS. 
            PERFORM C000-SUMMARY-REPORT. 
            PERFORM D000-TERMINATION. 
            STOP RUN. 
 



 

 

      / 
 
      ****************************************************************** 
      *                                                                * 
      *  A000 INITIALIZATION                                           * 
      *                                                                * 
      *    - OPENS THE INPUT AND OUTPUT FILES                          * 
      *    - WRITES THE FIRST PAGE COLUMN-HEADINGS 
      *    - READS FIRST RECORD                                        * 
      *                                                                * 
      ****************************************************************** 
 
       A000-INITIALIZATION. 
            PERFORM A100-OPEN-FILES. 
            PERFORM A200-GET-DATE. 
            PERFORM A300-PAGE-HEADING. 
            PERFORM X100-READ-RECORD. 
 
       A100-OPEN-FILES. 
            OPEN  INPUT CUSTOMER-FILE 
                  OUTPUT REPORT-FILE. 
 
       A200-GET-DATE. 
           ACCEPT WS10-CURRENT-DATE FROM DATE. 
           MOVE WS10-YEAR-IN   TO WS11-YEAR-OUT. 
           MOVE WS10-MONTH-IN  TO WS11-MONTH-OUT. 
           MOVE WS10-DAY-IN    TO WS11-MONTH-OUT. 
   
     A300-PAGE-HEADING. 
            WRITE REPORT-LINE FROM WS7-BLANK-LINE 
                 AFTER PAGE. 
            ADD 1 TO WS1-PAGE-NUM. 
            MOVE WS1-PAGE-NUM TO WS11-PAGE-NO-OUT. 
            WRITE REPORT-LINE FROM WS11-DATE-LINE 
                AFTER ADVANCING 2 LINES. 
            WRITE REPORT-LINE FROM WS5-TITLE 
                 AFTER ADVANCING 1 LINES. 
            WRITE REPORT-LINE FROM WS6-COLUMN-HEADINGS 
                 AFTER ADVANCING 3 LINES. 
 
      ****************************************************************** 
      *                                                                * 
      *  B000 CLUB CUSTOMER PROCESS                                    * 
      *                                                                * 
      *    THIS PARAGRAPH IS PERFORMED UNTIL THERE ARE NO MORE MASTER  * 
      *    RECORDS.  IT:                                               * 
      *                                                                * 
      *    - PERFORMS THE CALCULATION PARAGRAPH                        * 
      *    - PERFORMS THE DETAIL WRITING PARAGRAPH                     * 
      *    - PERFORMS THE READ PARAGRAPH                               * 
      *                                                                * 
      ****************************************************************** 
 
       B000-CLUB-CUSTOMER-PROCESS. 
            PERFORM B100-CALCULATION-PARA. 
            PERFORM B200-ROUTINE-WRITING-ANSWER. 
            PERFORM X100-READ-RECORD. 



 

 

 
      / 
      ****************************************************************** 
      *                                                                * 
      *  B100 CALCULATION-PARA.                                        * 
      *                                                                * 
      *    - CALCULATES THE BAR BILL DISCOUNTS                         * 
      *    - CALCULATES THE CUSTOMER'S TOTAL BILL AMOUNT               * 
      *    - ACCUMULATES THE GRAND TOTAL BILL AMOUNT                   * 
      *                                                                * 
      ****************************************************************** 
 
       B100-CALCULATION-PARA. 
            IF WS3-REST-BILL IS GREATER THAN 200 
               MULTIPLY 0.98 BY WS3-REST-BILL. 
            IF WS3-BAR-BILL > 400 
               MULTIPLY .95 BY WS3-BAR-BILL. 
            ADD WS3-REST-BILL, WS3-BAR-BILL, WS3-OTHER-CHARGES  
               GIVING WS1-TOTAL-BILL. 
            IF WS1-TOTAL-BILL GREATER THAN 1000 
               MULTIPLY WS1-TOTAL-BILL BY 0.97 GIVING WS1-TOTAL-BILL. 
            ADD WS3-AMT-OUTSTANDING TO WS1-TOTAL-BILL. 
            ADD WS1-TOTAL-BILL TO WS1-TOTAL-AMT-DUE. 
      /  
 
      ****************************************************************** 
      *                                                                * 
      *  B200 ROUTINE WRITING ANSWER                                   * 
      *                                                                * 
      *    - MOVES INPUT FIELDS TO OUTPUT LINE                         * 
      *    - MOVES CUSTOMER'S TOTAL BILL AMOUNT TO OUTPUT LINE         * 
      *    - WRITES DETAIL LINE                                        * 
      *                                                                * 
      ****************************************************************** 
 
       B200-ROUTINE-WRITING-ANSWER. 
            MOVE WS1-TOTAL-BILL TO WS4-AMT-DUE. 
 
      ****************************************************************** 
      *                                                                * 
      * NOTE:    WHEN TOTAL-BILL IS MOVED TO AMT-DUE, A  PROBLEM COULD * 
      * OCCUR  BECAUSE THE  PICTURE  FOR  TOTAL-BILL  IS ONE CHARACTER * 
      * LARGER THAN THE PICTURE  FOR  AMT-DUE  AND  YOU COULD LOSE THE * 
      * HIGH ORDER DIGIT OF THE LARGER PICTURE.                        * 
      *                                                                * 
      ****************************************************************** 
 
            MOVE WS3-MEMBER-NUMBER-IN TO WS4-MEMBER-NUMBER-OUT. 
            MOVE WS3-NAME-IN TO WS4-NAME-OUT. 
            MOVE WS3-MEMBER-ADDRESS-IN TO WS4-MEMBER-ADDRESS-OUT. 
            WRITE REPORT-LINE FROM WS4-DETAIL-LINE 
                AFTER ADVANCING 2 LINES. 
 
      / 
 
      ****************************************************************** 
      *                                                                * 



 

 

      *  C000 SUMMARY REPORT 
      *                                                                * 
      *    - MOVES GRAND TOTAL BILLING AMOUNT TO SUMMARY LINE          * 
      *    - WRITES GRAND TOTAL LINES                                  * 
      *                                                                * 
      ****************************************************************** 
 
       C000-SUMMARY-REPORT. 
            WRITE REPORT-LINE FROM WS9-STAR-LINE 
                AFTER ADVANCING 3 LINES. 
            MOVE WS1-TOTAL-AMT-DUE TO WS7-TOTAL-AMOUNT.  
 
      ****************************************************************** 
      *                                                                * 
      * NOTE: THE ABOVE MOVE STATEMENT IS A SIMPLE MOVE, WE ARE MOVING * 
      * DATA FROM  ONE  LOCATION  TO  ANOTHER  WITH  BOTH  DATA  NAMES * 
      * (TOTAL-AMT-DUE AMOUNT) BEING UNIQUE; THAT IS, THEY APPEAR ONLY * 
      * ONCE IN THE DATA DIVISION.                                     * 
      *                                                                * 
      ****************************************************************** 
 
            WRITE REPORT-LINE FROM WS7-TOTAL-LINE 
                AFTER ADVANCING 3 LINES. 
 
 
       D000-TERMINATION. 
            CLOSE CUSTOMER-FILE 
                  REPORT-FILE. 
      /   
 
       X100-READ-RECORD. 
            READ CUSTOMER-FILE INTO WS3-CUSTOMER-RECORD-IN 
                 AT END 
                      MOVE 'NO ' TO WS2-ARE-THERE-MORE-RECORDS. 
?? 
 
 
 
  
 
  
 
 
 


